NAG Library Routine Document

G05SRF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SRF generates a vector of pseudorandom numbers from a von Mises distribution with concentration parameter κ.

2 Specification

```
SUBROUTINE G05SRF(N, VK, STATE, X, IFAIL)
INTEGER              N, STATE(lstate), IFAIL
double precision     VK, X(N)
```

3 Description

The von Mises distribution is a symmetric distribution used in the analysis of circular data. The PDF (probability density function) of this distribution on the circle with mean direction $\mu_0 = 0$ and concentration parameter κ, can be written as:

$$f(\theta) = \frac{e^{\kappa \cos \theta}}{2\pi I_0(\kappa)},$$

where θ is reduced modulo 2π so that $-\pi \leq \theta < \pi$ and $\kappa \geq 0$. For very small κ the distribution is almost the uniform distribution, whereas for $\kappa \to \infty$ all the probability is concentrated at one point.

The n variates, $\theta_1, \theta_2, \ldots, \theta_n$, are generated using an envelope rejection method with a wrapped Cauchy target distribution as proposed by Best and Fisher (1979) and described by Dagpunar (1988).

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SRF.

4 References

Best D J and Fisher N I (1979) Efficient simulation of the von Mises distribution *Appl. Statist.* **28** 152–157

Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

Mardia K V (1972) *Statistics of Directional Data* Academic Press

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of pseudorandom numbers to be generated.

Constraint: N ≥ 0.

2: VK – ***double precision*** *Input*

On entry: κ, the concentration parameter of the required von Mises distribution.

Constraint: $0.0 <$ VK $\leq \sqrt{\text{X02ALF}}/2.0$.

3: STATE($lstate$) – INTEGER array *Communication Array*

Note: $lstate$ = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

4: X(N) – ***double precision*** array *Output*

On exit: the n pseudorandom numbers from the specified von Mises distribution.

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N < 0.

IFAIL $= 2$

On entry, VK ≤ 0.0,
or VK $> \sqrt{\text{X02ALF}()}/2.0$.

IFAIL $= 3$

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

For a given number of random variates the generation time increases slightly with increasing κ.

9 Example

This example prints the first five pseudorandom numbers from a von Mises distribution with $\kappa = 1.0$, generated by a single call to G05SRF, after initialization by G05KFF.

9.1 Program Text

```
*     G05SRF Example Program Text
*     Mark 22 Release. NAG Copyright 2006.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSTATE, MSEED, N
      PARAMETER        (MSTATE=633,MSEED=1,N=5)
*     .. Local Scalars ..
      DOUBLE PRECISION VK
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, SUBID
*     .. Local Arrays ..
      DOUBLE PRECISION X(N)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*     .. External Subroutines ..
      EXTERNAL         G05KFF, G05SRF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05SRF Example Program Results'
      WRITE (NOUT,*)
*     Set the distribution parameters VK
      VK = 1.0D0
*     Initialize the seed to a repeatable sequence
      SEED(1) = 1762543
*     GENID and SUBID identify the base generator.
      GENID = 1
      SUBID = 1
*     Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*     Generate the variates
      IFAIL = 1
      CALL G05SRF(N,VK,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*     Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
   20 CONTINUE
*
99999 FORMAT (1X,F10.4)
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05SRF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05SRF Example Program Results

     1.2947
    -1.9542
    -0.6464
    -1.4172
     1.2536
```